

## نکاتی در مورد استفاده از چاپگر

- ۱- چاپ محتویات یک RichTextBox
- ۲- چاپ محتویات کل یک فرم
- ۳- چاپ تصاویر
- ۴- چاپ یک خط
- ۵- چاپ متن
- ۶- تنظیم قلم (فونت) چاپگر
- ۷- چاپ متن در مکانی مشخص
- ۸- چاپ متن در مختصاتی مشخص
- ۹- چاپ متن تحت زاویه ای مشخص
- ۱۰- اگر چاپگر خواص قلم را رعایت نکند
- ۱۱- تنظیم مجدد چاپگر
- ۱۲- چاپ یک صفحه وب
- ۱۳- آیا چاپگر نصب شده است؟
- ۱۴- انتخاب چاپگر در زمان اجرای برنامه
- ۱۵- چاپ MSFlexGrid
- ۱۶- چاپ سریع چند کپی از یک سند
- ۱۷- تعیین حاشیه کاغذ چاپگر
- ۱۸- بدست آوردن اطلاعاتی در مورد چاپگر
- ۱۹- نمایش دادن فرم استاندارد چاپگرها
- ۲۰- تعیین جهت کاغذ چاپگر
- ۲۱- چگونه تغییرات اعمالی روی چاپگر را به فرم های برنامه اعمال کنیم؟
- ۲۲- کپی کردن اطلاعات یک تصویر، داخل تخته برش (clipboard) و چاپ آن

### ۱- چاپ محتویات یک RichTextBox :

برای اینکار یک دکمه و یک RichTextBox از منوی Project -> Components روی فرم قرار دهید. سپس در روال رخداد آن بنویسید :

```
Private Sub Command1_Click()  
    Call RichTextBox1.SelPrint(Printer.hdc)  
End Sub
```

البته لازم به ذکر است که این دستور تصاویر موجود در RichTextBox را چاپ نمی کند.

### ۲- چاپ محتویات کل یک فرم :

```
Private Sub Form_Load()  
    ' replace the "Form1" below with the name of the  
    ' form you want to print  
    Form1.PrintForm  
End Sub
```

### ۳- چاپ تصاویر :

یک دکمه و یک Picture Box روی فرم قرار دهید و در خاصیت تصویر آن یک تصویر قرار دهید. روی دکمه دوبار کلیک کنید و در روال رخداد آن بنویسید :

```
Private Sub Command1_Click()  
    Printer.PaintPicture Picture1.Picture, 0, 0, _  
        Picture1.Width, Picture1.Height  
    Printer.EndDoc  
End Sub
```

نکاتی در مورد کد فوق :

Picture1 را با نام هر کنترل دیگری که می خواهید تصویر موجود در آن را چاپ کنید عوض کنید. اعداد "0, 0" بیانگر محل چاپ شدن تصویر روی کاغذ می باشند. (بالا سمت چپ)، اگر می خواهید تصویری را بلافاصله پس از چاپ متنی چاپ کنید از مختصات "Printer.CurrentX, Printer.CurrentY" استفاده کنید. و اگر می خواهید تصویر چاپ شده را کوچک یا بزرگ نمایید "Picture1.Width, Picture1.Height" را با اعداد خود تعویض نمایید. برای مثال با "Picture1.Width \* 0.5, Picture1.Height \* 0.5" اندازه تصویر چاپ شده، نصف تصویر اصلی خواهد بود.

### ۴- چاپ یک خط :

یک دکمه روی فرم قرار دهید و ...

```
Public Sub PrintLine(Width As Single)  
    Printer.Line (0, Printer.CurrentY)-(Printer.ScaleWidth, _  
        Printer.CurrentY + Width), , BF  
    ' use the EndDoc command if the line is the last item you want  
    ' to print on the paper  
    Printer.EndDoc  
End Sub  
Private Sub Command1_Click()  
    ' replace the 40 below with the line width  
    PrintLine (40)  
End Sub
```

## ۵- چاپ متن :

یک دکمه روی فرم قرار دهید و ...

```
Private Sub Command1_Click()
    ' the following example will print hello on the form
    Printer.Print "hello"
    ' use the EndDoc command if this text is the last thing you want
    ' to print on the paper
    Printer.EndDoc
End Sub
```

## ۶- تنظیم قلم (فونت) چاپگر:

یک دکمه روی فرم قرار دهید و ...

```
Private Sub Command1_Click()
    Printer.FontName = "Arial"
    Printer.FontUnderline = False
    Printer.FontBold = False
    Printer.FontItalic = True
    Printer.FontSize = "30"
    Printer.Print "hello"
    ' use the EndDoc command if this text is the last
    ' thing you want to print on the paper.
    Printer.EndDoc
End Sub
```

## ۷- چاپ متن در مکانی مشخص :

یک دکمه روی فرم قرار دهید و ...

```
Public Sub PrintAlignedText(s As String, Alignment As String)
    Select Case Alignment
    Case "Center"
        Printer.CurrentX = (Printer.ScaleWidth - _
            Printer.TextWidth(s)) \ 2
    Case "Left"
        Printer.CurrentX = 0
    Case "Right"
        Printer.CurrentX = Printer.ScaleWidth - _
            Printer.TextWidth(s)
    End Select
    Printer.Print s
    ' use the EndDoc command if this text is the last
    ' thing you want to print on the paper.
    Printer.EndDoc
End Sub
Private Sub Command1_Click()
    Call PrintAlignedText("hello", "Center")
End Sub
```

## ۸- چاپ متن در مختصات مشخص :

یک دکمه روی فرم قرار دهید و ...

```

Private Sub Command1_Click()
' this example will print "hello" in the coordinates 100, 300
  Printer.CurrentX = 100
  Printer.CurrentY = 300
  Printer.Print "hello"
' use the EndDoc command if this text is the last thing you want
' to print on the paper
  Printer.EndDoc
End Sub

```

## ۹- چاپ متن تحت زاویه ای مشخص :

یک دکمه روی فرم قرار دهید و یک ماژول جدید به برنامه اضافه کنید سپس کد زیر را در آن بنویسید :

```

Option Explicit
Public Const LF_FACESIZE = 32
Public Type LOGFONT
  lfHeight As Long
  lfWidth As Long
  lfEscapement As Long
  lfOrientation As Long
  lfWeight As Long
  lfItalic As Byte
  lfUnderline As Byte
  lfStrikeOut As Byte
  lfCharSet As Byte
  lfOutPrecision As Byte
  lfClipPrecision As Byte
  lfQuality As Byte
  lfPitchAndFamily As Byte
  lfFaceName As String * LF_FACESIZE
End Type
Public Type DOCINFO
  cbSize As Long
  lpszDocName As String
  lpszOutput As String
  lpszDatatype As String
  fwType As Long
End Type
Declare Function CreateFontIndirect Lib "gdi32" Alias _
  "CreateFontIndirectA" (lpLogFont As LOGFONT) As Long

Declare Function SelectObject Lib "gdi32" _
  (ByVal hdc As Long, ByVal hObject As Long) As Long

Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) _
  As Long

Declare Function CreateDC Lib "gdi32" Alias "CreateDCA" _
  (ByVal lpDriverName As String, ByVal lpDeviceName As String, _
  ByVal lpOutput As Long, ByVal lpInitData As Long) As Long

Declare Function DeleteDC Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function TextOut Lib "gdi32" Alias "TextOutA" _
  (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, _
  ByVal lpString As String, ByVal nCount As Long) As Long ' or
' Boolean

```

```

Declare Function StartDoc Lib "gdi32" Alias "StartDocA" _
    (ByVal hdc As Long, lpdi As DOCINFO) As Long

Declare Function EndDoc Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function StartPage Lib "gdi32" (ByVal hdc As Long) As Long

Declare Function EndPage Lib "gdi32" (ByVal hdc As Long) As Long

```

روی دکمه دوبار کلیک کنید و در روال رخداده آن بنویسید :

```

Private Sub Command1_Click()
    Const DESIREDFONTSIZE = 12 ' Font Size.

    Dim OutString As String 'String to be rotated
    Dim lf As LOGFONT 'Structure for setting up rotated font
    Dim temp As String 'Temp string var
    Dim result As Long 'Return value for calling API functions
    Dim hOldfont As Long 'Hold old font information
    Dim hPrintDc As Long 'Handle to printer dc
    Dim hFont As Long 'Handle to new Font
    Dim di As DOCINFO 'Structure for Print Document info

    OutString = "Hello World" 'Set string to be rotated

    'Set rotation in tenths of a degree, i.e., 1800 = 180 degrees
    lf.lfEscapement = 1800
    lf.lfHeight = (DESIREDFONTSIZE * -60) / Screen.TwipsPerPixelY
    hFont = CreateFontIndirect(lf) 'Create the rotated font
    di.cbSize = 20 'Size of DOCINFO structure
    di.lpszDocName = "My Document" 'Set name of print job(Optional)

    'Create a printer device context
    hPrintDc = CreateDC(Printer.DriverName, Printer.DeviceName, 0, 0)

    result = StartDoc(hPrintDc, di) 'Start a new print document
    result = StartPage(hPrintDc) 'Start a new page

    'Select our rotated font structure and save previous font info
    hOldfont = SelectObject(hPrintDc, hFont)

    'Send rotated text to printer, starting at location 1000, 1000
    result = TextOut(hPrintDc, 1000, 1000, OutString, Len(OutString))

    'Reset font back to original, non-rotated
    result = SelectObject(hPrintDc, hOldfont)

    'Send non-rotated text to printer at same page location
    result = TextOut(hPrintDc, 1000, 1000, OutString, Len(OutString))

    result = EndPage(hPrintDc) 'End the page
    result = EndDoc(hPrintDc) 'End the print job
    result = DeleteDC(hPrintDc) 'Delete the printer device context
    result = DeleteObject(hFont) 'Delete the font object
End Sub

```

۱۰- اگر چاپگر خواص قلم را رعایت نکند :

در این حالت بجای استفاده از [font name] Printer.font =

از کد زیر استفاده کنید :

```
Dim F as New StdFont
F.Name = [font name]
F.Size = [size]
Set Printer.Font = F
```

۱۱- تنظیم مجدد چاپگر :

```
Printer.Print "" 'Reset printer
Printer.Font = "" 'used to reset font details.
```

۱۲- چاپ یک صفحه وب :

```
Public Sub PrintWebPage(URL As String)
    Shell "rundll32.exe c:\windows\system\MSHTML.dll,PrintHTML " & _
        URL, vbNormalFocus
End Sub
```

مثال :

```
PrintWebPage "http://www.vbworld.com/"
```

۱۳- آیا چاپگر نصب شده است ؟

```
Public Function PrinterInstalled() As Boolean
    On Error Resume Next
    Dim strDummy As String
    strDummy = Printer.DeviceName
    If Err.Number Then
        PrinterInstalled = False
    Else
        PrinterInstalled = True
    End If
End Function
```

۱۴- انتخاب چاپگر در زمان اجرای برنامه :

```
'Option Explicit
Private Sub Form_Load()
Dim P As Printer
' Display the current default printer name and port.
cboPrinters.Text = Printer.DeviceName
lblPort.Caption = Printer.Port
' Loop through the printers collection and check each printer name.
For Each P In Printers
' Skip any printers you don't want to appear in the combo box.
If P.DeviceName = "Rendering Subsystem" Then
'Don't add it to the combo box.
Else
' Add any printers you do want to appear in the combo box.
cboPrinters.AddItem P.DeviceName
End If
Next
End Sub
Private Sub cboPrinters_Click()
Dim P As Printer
' Loop through the printers collection and check each printer name.
```

```

For Each P In Printers
    ' If the printers collection name matches the printer selected in
    the
    ' combo box, display it's port.
    If P.DeviceName = cboPrinters.List(cboPrinters.ListIndex) Then
        lblPort.Caption = P.Port
        Exit For
    End If
End If
Next
End Sub
Private Sub cmdOK_Click()
Dim P As Printer
' Loop through the printers collection and check each printer name.
For Each P In Printers
    ' If the printers collection name is the same as the printer
    selected
    ' in the combo box, set that printer as the applications default
    printer.
    If P.DeviceName = cboPrinters.List(cboPrinters.ListIndex) Then
        Set Printer = P
        Exit For
    End If
Next
' Optional message box.
MsgBox "All printing for this application will be sent to the " & _
        Printer.DeviceName & " printer on " & Printer.Port, _
        vbOKOnly, "Print"
Unload Me
End Sub
Private Sub cmdCancel_Click()
Unload Me
End Sub

```

## ۱۵- چاپ MSFlexGrid :

برای چاپ محتویات آن می توان از دستور زیر استفاده کرد :

```

Printer.PaintPicture MSFlexGrid_Name.Picture, 0, 0
Printer.EndDoc

```

و اگر می خواهید که کل صفحه کاغذ چاپگر را ببوشاند قبل از دو خط فوق بنویسید :

```

Dim old_width as Integer
MSFlexGrid_Name.width=printer.width

```

راه دوم :

```

Private Sub PrintFlexGrid(ByVal ptr As Object, _
        ByVal flx As MSFlexGrid, ByVal xmin As Single, _
        ByVal ymin As Single)

```

```

Const GAP = 60

```

```

Dim xmax As Single
Dim ymax As Single
Dim X As Single
Dim c As Integer
Dim r As Integer

```

```

With flxData
    ' See how wide the whole thing is.
    xmax = xmin + GAP
    For c = 0 To .Cols - 1
        xmax = xmax + .ColWidth(c) + 2 * GAP
    Next c

    ' Print each row.
    ptr.CurrentY = ymin
    For r = 0 To .Rows - 1
        ' Draw a line above this row.
        If r > 0 Then ptr.Line (xmin, ptr.CurrentY)-(xmax, _
            ptr.CurrentY)
        ptr.CurrentY = ptr.CurrentY + GAP

        ' Print the entries on this row.
        X = xmin + GAP
        For c = 0 To .Cols - 1
            ptr.CurrentX = X
            ptr.Print BoundedText(.TextMatrix(r, c), _
                .ColWidth(c));
            X = X + .ColWidth(c) + 2 * GAP
        Next c
        ptr.CurrentY = ptr.CurrentY + GAP

        ' Move to the next line.
        ptr.Print
    Next r
    ymax = ptr.CurrentY

    ' Draw a box around everything.
    ptr.Line (xmin, ymin)-(xmax, ymax), , B

    ' Draw lines between the columns.
    X = xmin
    For c = 0 To .Cols - 2
        X = X + .ColWidth(c) + 2 * GAP
        ptr.Line (X, ymin)-(X, ymax)
    Next c
End With
End Sub

' Truncate the string so it fits within the width.
Private Function BoundedText(ByVal ptr As Object, _
    ByVal txt As String, ByVal max_wid As Single) As String
    Do While ptr.TextWidth(txt) > max_wid
        txt = Left$(txt, Len(txt) - 1)
    Loop
    BoundedText = txt
End Function

```

مثال :

```

Private Sub cmdPrint_Click()
    PrintFlexGrid Printer, flxData, 1440, 1440
    Printer.EndDoc
End Sub

```

و در مثال زیر یک پیش نمایش فرم چاپی روی Form2 ایجاد خواهد شد :

```

Private Sub cmdPreview_Click()
    Form2.AutoRedraw = True
    PrintFlexGrid Form2, flxData, 500, 500
    Form2.Show

```



End Sub

## ۱۶- چاپ سریع چند کپی از یک سند:

```
For i = 1 To num_copies
    ' Print the document here...
    Printer.NewPage
Next i
Printer.EndDoc
```

## ۱۷- تعیین حاشیه کاغذ چاپگر:

```
Printer.ScaleLeft = -0.5 * 1440
Printer.ScaleTop = -0.75 * 1440
Printer.CurrentX = 0
Printer.CurrentY = 0
```

دستورات فوق حاشیه بالای کاغذ را برابر ۰/۷۵ اینچ و حاشیه سمت چپ کاغذ را برابر ۰/۵ اینچ قرار می دهند. ضریب ۱۴۴۰ اینچ را به twips تبدیل می کند.

## ۱۸- بدست آوردن اطلاعاتی در مورد چاپگر:

یک دکمه روی فرم قرار دهید و یک ماژول جدید به برنامه اضافه کنید سپس کد زیر را در آن بنویسید:

```
Public Declare Function GetProfileString Lib "kernel32" _
Alias "GetProfileStringA" ( _
    ByVal lpAppName As String, _
    ByVal lpKeyName As String, _
    ByVal lpDefault As String, _
    ByVal lpReturnedString As String, _
    ByVal nSize As Long) As Long
```

روی دکمه دوبار کلیک کنید و در روال رخداده آن بنویسید:

```
Private Sub Command1_Click()
    'String variables to store section and key
    AppName$ = "windows"
    KeyName$ = "device"
    ' Return string length
    nSize% = 81
    RetStr$ = Space$(nSize%)
    NumChars% = GetProfileString(AppName$, KeyName$, NullStr$, RetStr$,
    nSize%)
    ' Store the string
    koRetStr$ = Left$(RetStr$, NumChars%)
    ' Parse the string for specifics
    ' Find the first comma
    CommaPos1% = InStr(1, RetStr$, ",")
    ' Find the next comma
    CommaPos2% = InStr(CommaPos1% + 1, RetStr$, ",")
    ' Get Windows printer type
    PrinterCaption = Left$(RetStr$, CommaPos1% - 1)
    'Get the Printer driver
    PrinterDriver = Mid$(RetStr$, CommaPos1% + 1, _
        CommaPos2% - CommaPos1% - 1) & ".DRV"
    'Get the Printer port
```

```
Printerport = Mid$(RetStr$, CommaPos2% + 1)

MsgBox "Windows printer type = " + PrinterCaption + vbCrLf + _
      "Printer driver = " + PrinterDriver + vbCrLf + _
      "Printer port = " + Printerport, vbInformation, _
      "Printer"
End Sub
```

## ۱۹- نمایش دادن فرم استاندارد چاپگرها :

یک دکمه و کنترل Common Dialog را از منوی Project -> Components... روی فرم قرار دهید و ...

```
Private Sub cmdPrnSetup_Click ()
    CMDialog1.Flags = &H40 ' force the printers setup dialog box
    CMDialog1.Action = 5 ' run the printer setup
End Sub
```

## ۲۰- تعیین جهت کاغذ چاپگر :

```
Public Declare Function Escape Lib "gdi32" Alias "Escape" (
    ByVal hdc As Long, ByVal nEscape As Long, ByVal nCount As Long, _
    ByVal lpInData As String, lpOutData As Any) As Long
```

```
Sub Command1_Click ()
    Const PORTRAIT = 1
    Const LANDSCAPE = 2
    Const GETSETPAPERORIENT = 30
    Dim Orient As OrientStructure
    Printer.Print ""
    Orient.Orientation = LANDSCAPE
    x% = Escape(Printer.hDC, GETSETPAPERORIENT, Len(Orient) _
        , "", Null)
    Print x%
End Sub
```

## ۲۱- چگونه تغییرات اعمالی روی چاپگر را به فرم های برنامه اعمال کنیم ؟

هنگامیکه برنامه بارگذاری می شود اطلاعات مربوط به چاپگر را نیز بارگذاری می کند. اگر هنگام اجرای برنامه تغییراتی روی نوع یا مشخصات چاپگر اعمال شود تا قطعه کد زیر اجرا نگردد برنامه به روز در نخواهد آمد.

```
Private Declare Function SendMessage Lib "user32" Alias _
    "SendMessageA" (ByVal hwnd As Long, _
    ByVal wParam As Long, ByVal lParam As Long, _
    lParam As Any) As Long
```

```
Private Sub Command1_Click()

    Const WM_WININICHANGE = &H1A
    Const HWND_BROADCAST = &HFFFFFF
    ' Dummy means send to all top windows.
    ' Send name of changed section as lParam.
    lRes = SendMessage(HWND_BROADCAST, WM_WININICHANGE, 0, _
        ByVal "Windows")
End Sub
```

۲۲- کپی کردن اطلاعات یک تصویر، داخل تخته برش (clipboard) و چاپ آن :  
گاهی اوقات خاصیت PrintForm شی چاپگر بعضی از کنترل های روی فرم را مستثنی می سازد. کد زیر فشرده شدن دکمه PrintScrn را شبیه سازی می کند، با این کار اطلاعات داخل تخته برش کپی می شوند و سپس می توان آن را چاپ کرد.

```
Private Declare Sub keybd_event Lib "user32" (ByVal bVk As Byte, _
    ByVal bScan As Byte, ByVal dwFlags As Long, _
    ByVal dwExtraInfo As Long)
Const VK_MENU As Byte = &H12
Const VK_SNAPSHOT As Byte = &H2C
Const KEYEVENTF_KEYUP = &H2

Private Sub cmdPrintForm_Click()
    Dim lWidth As Long, lHeight As Long
    Clipboard.Clear
    Call keybd_event(VK_MENU, 0, 0, 0)
    Call keybd_event(VK_SNAPSHOT, 0, 0, 0)
    DoEvents
    Call keybd_event(VK_SNAPSHOT, 0, KEYEVENTF_KEYUP, 0)
    Call keybd_event(VK_MENU, 0, KEYEVENTF_KEYUP, 0)
    Printer.Print
    If Width > Printer.ScaleWidth Then
        lWidth = Printer.ScaleWidth
        lHeight = (Printer.ScaleWidth / Width) * Height
    Else
        lWidth = Width
        lHeight = Height
    End If
    Printer.PaintPicture Clipboard.GetData, 0, 0, lWidth, lHeight
    Printer.EndDoc
    ' SavePicture Clipboard.GetData, Me.Name & ".BMP"
End Sub

Private Sub cmdPrintScreen_Click()
    Dim lHeight As Long
    Clipboard.Clear
    Call keybd_event(VK_SNAPSHOT, 1, 0, 0)
    DoEvents
    Call keybd_event(VK_SNAPSHOT, 1, KEYEVENTF_KEYUP, 0)
    Printer.Print
    lHeight = (Printer.ScaleWidth / Screen.Width) * Screen.Height
    Printer.PaintPicture Clipboard.GetData, 0, 0, Printer.ScaleWidth,
lHeight
    Printer.EndDoc
    ' SavePicture Clipboard.GetData, Me.Name & ".BMP"
End Sub
```